NAG Fortran Library Routine Document

F07JRF (ZPTTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07JRF (ZPTTRF) computes the modified Cholesky factorization of a complex n by n Hermitian positive-definite tridiagonal matrix A.

2 Specification

```
SUBROUTINE F07JRF (N, D, E, INFO)
INTEGER              N, INFO
double precision     D(*)
complex*16           E(*)
```

The routine may be called by its LAPACK name ***zpttrf***.

3 Description

F07JRF (ZPTTRF) factorizes the matrix A as

$$A = LDL^{\mathrm{H}},$$

where L is a unit lower bidiagonal matrix and D is a diagonal matrix with positive diagonal elements. The factorization may also be regarded as having the form $U^{\mathrm{H}}DU$, where U is a unit upper bidiagonal matrix.

4 References

None.

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: D($*$) – ***double precision*** array *Input/Output*

Note: the dimension of the array D must be at least $\max(1, \mathrm{N})$.

On entry: must contain the n diagonal elements of the matrix A.

On exit: is overwritten by the n diagonal elements of the diagonal matrix D from the LDL^{H} factorization of A.

3: E($*$) – ***complex*16*** array *Input/Output*

Note: the dimension of the array E must be at least $\max(1, \mathrm{N} - 1)$.

On entry: must contain either the $(n-1)$ subdiagonal elements, or the $(n-1)$ superdiagonal elements, of the matrix A.

On exit: if the subdiagonal elements of A were supplied, then E is overwritten by the $(n-1)$ subdiagonal elements of the lower bidiagonal matrix L.

If the superdiagonal elements of A were supplied, then E overwritten by the $(n-1)$ superdiagonal elements of the upper bidiagonal matrix U.

4: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the leading minor of order i is not positive-definite. If $i <$ N the factorization could not be completed, while if $i =$ N, the factorization was completed, but D(N) ≤ 0.

7 Accuracy

The computed factorization satisfies an equation of the form

$$A + E = LDL^{\mathrm{H}},$$

where

$$\|E\|_\infty = O(\epsilon)\|A\|_\infty$$

and ϵ is the ***machine precision***.

Following the use of this routine, F07JSF (ZPTTRS) can be used to solve systems of equations $AX = B$, and F07JUF (ZPTCON) can be used to estimate the condition number of A.

8 Further Comments

The total number of floating-point operations required to factorize the matrix A is proportional to n.

The real analogue of this routine is F07JDF (DPTTRF).

9 Example

To factorize the Hermitian positive-definite tridiagonal matrix A given by

$$A = \begin{pmatrix} 16.0 & 16.0 - 16.0i & 0 & 0 \\ 16.0 + 16.0i & 41.0 & 18.0 + 9.0i & 0 \\ 0 & 18.0 - 9.0i & 46.0 & 1.0 + 4.0i \\ 0 & 0 & 1.0 - 4.0i & 21.0 \end{pmatrix}.$$

9.1 Program Text

```
*     F07JRF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=50)
*     .. Local Scalars ..
      INTEGER          I, INFO, N
*     .. Local Arrays ..
      COMPLEX *16      E(NMAX-1)
      DOUBLE PRECISION D(NMAX)
```

```
*     .. External Subroutines ..
      EXTERNAL         ZPTTRF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07JRF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the lower bidiagonal part of the tridiagonal matrix A from
*        data file
*
         READ (NIN,*) (D(I),I=1,N)
         READ (NIN,*) (E(I),I=1,N-1)
*
*        Factorize the tridiagonal matrix A
*
         CALL ZPTTRF(N,D,E,INFO)
*
         IF (INFO.GT.0) THEN
            WRITE (NOUT,99999) 'The leading minor of order ', INFO,
     +        ' is not positive definite'
         END IF
*
*        Print details of the factorization
*
         WRITE (NOUT,*) 'Details of factorization'
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' The diagonal elements of D'
         WRITE (NOUT,99998) (D(I),I=1,N)
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     ' Sub-diagonal elements of the Cholesky factor L'
         WRITE (NOUT,99998) (E(I),I=1,N-1)
*
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I3,A)
99998 FORMAT (1X,8F9.4)
99997 FORMAT (1X,5I9)
      END
```

9.2 Program Data

```
F07JRF Example Program Data
    4                                                    :Value of N
   16.0          41.0          46.0          21.0        :End of diagonal D
 ( 16.0, 16.0) ( 18.0, -9.0) (  1.0, -4.0)               :End of sub-diagonal E
```

9.3 Program Results

```
 F07JRF Example Program Results

 Details of factorization

  The diagonal elements of D
   16.0000   9.0000   1.0000   4.0000

  Sub-diagonal elements of the Cholesky factor L
    1.0000   1.0000   2.0000  -1.0000   1.0000  -4.0000
```